
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 11.1.5.2 which is part of the UTRAN test suite. Only essential changes to the TTCN are applied and documented in section 4.

With these changes applied the test case can be demonstrated to run with one or more 3G UEs 
(see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Case:
TC_11_1_5_2
Test Group:
/11/UTRAN34_NISPC.ttcn
ATS Version:
iwd-TTCN3-B2013-03_D14wk37 + essential modifications.

System Simulator used:
Rohde & Schwarz CMW500

UE used:
MediaTek K2 Handset
Verification Status:
PASS
4 Corrections required for test case 11.1.5.2
4.1 Introduction

This section describes the changes required to make test case 11.1.5.2 run correctly with a 3G UE. All modifications are marked with label “WA#” in the TTCN comments column of the enclosed ATS [1].

The ATS version used as basis was UTRAN testsuite which is part of iwd-TTCN3-B2013-03_D14wk37 release. This testsuite provided by MCC160 contains UTRAN test cases.
4.2 template cs_RB_ActTimeInfoList20_22
	Record Name
	cs_RB_ActTimeInfoList20_22

	Reason for change
	In the function fl_TC_11_1_5_2_Body()->f_UTRAN34_SecPDP_Context_MT_r7-> f_UTRAN34_RB_SetUpDCH_64k_2AM_PS_RAB_r7-> f_UTRAN34_RRC_ReceiveRB_SetupCmpl_PS_RAB-> f_UTRAN34_SS_CipheringAM_RAB_UL_DL for the case cell_DCH_2AM_PS, the assignment v_RB_ActivationTimeInfoList := cs_RB_ActTimeInfoList20_22(v_RLC_SNs, v_RLC_SNs) is being done. 
Later in the same function the function f_UTRAN_ActivateCiphering_RLC_UL is called in which v_RB_ActivationTimeInfoList is being sent as a parameter. 
The problem is that the template cs_RB_ActTimeInfoList20_22 in the current implementation includes RB20 and RB 22 both. This results in the RB 20 being ciphered again. This is not required, since RB 20 is already ciphered at this point in the testcase.
This needs to be corrected.

	Summary of change
	Corrected the implementation of cs_RB_ActTimeInfoList20_22, please see the screenshot below.

	Source of change
	UTRAN34_RRC_Templates.ttcn

	
	


Before:

	template (value) RB_ActivationTimeInfoList cs_RB_ActTimeInfoList20_22(RLC_SequenceNumber p_RLC_SN20,
                                                                        RLC_SequenceNumber p_RLC_SN22) :=
  { /* @status    APPROVED (UTRAN) */
    cs_RB_ActTimeInfo(tsc_RB20, p_RLC_SN20),

    cs_RB_ActTimeInfo(tsc_RB22, p_RLC_SN22)
  };


After:

	template (value) RB_ActivationTimeInfoList cs_RB_ActTimeInfoList20_22(RLC_SequenceNumber p_RLC_SN20,
                                                                        RLC_SequenceNumber p_RLC_SN22) :=
  { /* @status    APPROVED (UTRAN) */
    //WA#11_1_5_2 cs_RB_ActTimeInfo(tsc_RB20, p_RLC_SN20),
    cs_RB_ActTimeInfo(tsc_RB22, p_RLC_SN22)
  };


4.3 Function f_UTRAN34_SecPDP_Context_MT_r7
	Record Name
	f_UTRAN34_SecPDP_Context_MT_r7

	Reason for change
	In the current TTCN implementation of function f_UTRAN34_SecPDP_Context_MT_r7 the QoS parameters that are being sent in the Activate Secondary PDP Context Accept message are not being set based on what was received in the Activate Secondary PDP Context REQ. This needs to be corrected. 

	Summary of change
	Assigned the variable v_Recd_QoS with the received QoS in the Activate Secondary PDP Context REQ. Also, in the Activate Secondary PDP Context Accept message, replaced p_QualityOfService.delayClass with v_Recd_QoS.delayClass, replaced p_QualityOfService.trafficClass with v_Recd_QoS.trafficClass and returned v_Recd_QoS instead of p_QualityOfService in the last line. 

Please see screenshot. 

	Source of change
	UTRAN34_RRC_Templates.ttcn

	
	


Before:

	function f_UTRAN34_SecPDP_Context_MT_r7(UTRAN_CellId_Type p_CellId,
                                          template (value) TrafficFlowTemplate p_TrafficFlowTemplate,
                                          PDPContextParameters p_FirstPDPContextParameters,
                                          template (value) QualityOfService p_QualityOfService,
                                          charstring p_Msg)
    runs on UTRAN_PTC return PDPContextParameters
  {
    var template (value) LLC_SAPI v_LLC_SAPI := f_UTRAN34_LLC_SAPI_TX();
    var TI v_TI_S := p_FirstPDPContextParameters.send_TI;
    var TI v_TI_R := p_FirstPDPContextParameters.received_TI;
    var IP_VersionType v_IP_VersionType := p_FirstPDPContextParameters.ip_VersionType;
    var NSAPI v_RecdNSAPI_2d;
    var NSAPI v_NSAPI := p_FirstPDPContextParameters.recdNSAPI;
    var RRC_DATA_IND v_RRC_DATA_IND;
    var TI v_TI_2_S;
    var TI v_TI_2_R;
    var template (value) QualityOfService  v_Recd_QoS := p_QualityOfService;  //@sic R5s140558 sic@
    timer t_T3885 := 8800.0;
    v_Recd_QoS.precedenceClass := '011'B; //Acc to 34.123-3 cl 8.10 @sic R5s140558 sic@
    //Set the value for TI for a secondary PDP context
    v_TI_2_S.tiFlag := '0'B;
    v_TI_2_S.tiVal := int2bit(bit2int(v_TI_S.tiVal) + 1, 3);
    //Request Secondary PDP Context activation
    U_Dc.send(cas_DataReq_PS(utran_CellDedicated,
                             tsc_RB3,
                             cs_U_SecPDPCOntextActReq(cs_ReqSecPDP_ContextActivation_MT(v_TI_2_S,
                                                                                        v_Recd_QoS,  //@sic R5s140558 sic@
                                                                                        cr_LinkedTI(v_TI_S),  //@sic R5s140387 sic@
                                                                                        p_TrafficFlowTemplate))));
    //Activate PDP Context Accept
    v_TI_2_R := v_TI_2_S;
    v_TI_2_R.tiFlag :='1'B;
    t_T3885.start;
    alt {
      [] t_T3885.timeout
        {
          f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, p_Msg);
        }
      [] U_Dc.receive(car_PS_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_ActSecPDPContReq( cr_ActSecPDP_ContextRequest(
                                                                                                            v_TI_2_R,
                                                                                                            cdr_QoS_InteractiveOrBackground_PrecedenceClass(p_QualityOfService.delayClass, p_QualityOfService.trafficClass), //@sic R5s140558 sic@
                                                                                                            cr_LinkedTI(v_TI_R)))))
                                                                                                     -> value v_RRC_DATA_IND
        {
          t_T3885.stop;
          v_RecdNSAPI_2d := v_RRC_DATA_IND.msg.activateSecondaryPDPContextRequestul.requestedNSAPI;         
          if (v_RecdNSAPI_2d != v_NSAPI) {
            f_UTRAN_PreliminaryPass(__FILE__, __LINE__, p_Msg);
          } else{
            f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, p_Msg);
          }
        }
    }
    f_UTRAN34_RB_SetUpDCH_64k_2AM_PS_RAB_r7(p_CellId, int2bit(bit2int(v_RecdNSAPI_2d.nSAPI_Value), 8));
    //Secondary PDP Context Accept
    U_Dc.send(cas_DataReq_PS(utran_CellDedicated,
                             tsc_RB3,
                             cs_U_SecPDPCOntextActAcpt(cs_ActSecPDP_ContextAcp_MT(v_TI_2_S,
                                                                                  v_LLC_SAPI,                                                                                  
p_QualityOfService.delayClass,
p_QualityOfService.trafficClass))));
    return valueof(cs_PDPContextParameters(v_TI_2_R, v_TI_2_S, v_IP_VersionType, -, v_RecdNSAPI_2d, p_QualityOfService));
  }


After:

	function f_UTRAN34_SecPDP_Context_MT_r7(UTRAN_CellId_Type p_CellId,
                                          template (value) TrafficFlowTemplate p_TrafficFlowTemplate,
                                          PDPContextParameters p_FirstPDPContextParameters,
                                          template (value) QualityOfService p_QualityOfService,
                                          charstring p_Msg)
    runs on UTRAN_PTC return PDPContextParameters
  {
    var template (value) LLC_SAPI v_LLC_SAPI := f_UTRAN34_LLC_SAPI_TX();
    var TI v_TI_S := p_FirstPDPContextParameters.send_TI;
    var TI v_TI_R := p_FirstPDPContextParameters.received_TI;
    var IP_VersionType v_IP_VersionType := p_FirstPDPContextParameters.ip_VersionType;
    var NSAPI v_RecdNSAPI_2d;
    var NSAPI v_NSAPI := p_FirstPDPContextParameters.recdNSAPI;
    var RRC_DATA_IND v_RRC_DATA_IND;
    var TI v_TI_2_S;
    var TI v_TI_2_R;
    var template (value) QualityOfService  v_Recd_QoS := p_QualityOfService;  //@sic R5s140558 sic@
    timer t_T3885 := 8800.0;
    v_Recd_QoS.precedenceClass := '011'B; //Acc to 34.123-3 cl 8.10 @sic R5s140558 sic@
    //Set the value for TI for a secondary PDP context
    v_TI_2_S.tiFlag := '0'B;
    v_TI_2_S.tiVal := int2bit(bit2int(v_TI_S.tiVal) + 1, 3);
    //Request Secondary PDP Context activation
    U_Dc.send(cas_DataReq_PS(utran_CellDedicated,
                             tsc_RB3,
                             cs_U_SecPDPCOntextActReq(cs_ReqSecPDP_ContextActivation_MT(v_TI_2_S,
                                                                                        v_Recd_QoS,  //@sic R5s140558 sic@
                                                                                        cr_LinkedTI(v_TI_S),  //@sic R5s140387 sic@
                                                                                        p_TrafficFlowTemplate))));
    //Activate PDP Context Accept
    v_TI_2_R := v_TI_2_S;
    v_TI_2_R.tiFlag :='1'B;
    t_T3885.start;
    alt {
      [] t_T3885.timeout
        {
          f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, p_Msg);
        }
      [] U_Dc.receive(car_PS_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_ActSecPDPContReq( cr_ActSecPDP_ContextRequest(
                                                                                                            v_TI_2_R,
                                                                                                            cdr_QoS_InteractiveOrBackground_PrecedenceClass(p_QualityOfService.delayClass, p_QualityOfService.trafficClass), //@sic R5s140558 sic@
                                                                                                            cr_LinkedTI(v_TI_R)))))
                                                                                                     -> value v_RRC_DATA_IND
        {
          t_T3885.stop;
          v_RecdNSAPI_2d := v_RRC_DATA_IND.msg.activateSecondaryPDPContextRequestul.requestedNSAPI;
          v_Recd_QoS := v_RRC_DATA_IND.msg.activateSecondaryPDPContextRequestul.requestedQoS;//WA#NISPC_Generic : This is required to capture the QoS sent by the UE
          if (v_RecdNSAPI_2d != v_NSAPI) {
            f_UTRAN_PreliminaryPass(__FILE__, __LINE__, p_Msg);
          } else{
            f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, p_Msg);
          }
        }
    }
    f_UTRAN34_RB_SetUpDCH_64k_2AM_PS_RAB_r7(p_CellId, int2bit(bit2int(v_RecdNSAPI_2d.nSAPI_Value), 8));
    //Secondary PDP Context Accept
    U_Dc.send(cas_DataReq_PS(utran_CellDedicated,
                             tsc_RB3,
                             cs_U_SecPDPCOntextActAcpt(cs_ActSecPDP_ContextAcp_MT(v_TI_2_S,
                                                                                  v_LLC_SAPI,                                                                                  
                                                                                  v_Recd_QoS.delayClass,/*WA#NISPC_Generic : replaced p_QualityOfService.delayClass*/
                                                                                  v_Recd_QoS.trafficClass/*WA#NISPC_Generic : replaced p_QualityOfService.trafficClass*/))));
    return valueof(cs_PDPContextParameters(v_TI_2_R, v_TI_2_S, v_IP_VersionType, -, v_RecdNSAPI_2d, v_Recd_QoS /*WA#NISPC_Generic : replaced p_QualityOfService*/));
  }


4.4 Function f_UTRAN34_SendReceiveData
	Record Name
	f_UTRAN34_SendReceiveData

	Reason for change
	The purpose of function f_UTRAN34_SendReceiveData is to send and receive IP loopback data. However, the current TTCN implementation sends and receives RLC Test Data when PDCP is configured. This needs to be corrected.

	Summary of change
	Corrected the implementation of function f_UTRAN34_SendReceiveData  to send and receive IP loopback data. Also created templates for PDCP Data handling. please see the screenshots below.

	Source of change
	UTRAN34_NISPC_Functions.ttcn

	
	


Before:

	function f_UTRAN34_SendReceiveData(SS_RB_Identity p_SendRB,
                                     SS_RB_Identity p_ReceiveRB,
                                     bitstring p_Data,
                                     boolean p_Loopback := true) runs on UTRAN_PTC
  {
    timer t_Dly := 1.0; //1s
    U_AM.send(cas_RLC_AM_TestData_Req(cs_RLC_AM_DataReq(utran_CellDedicated, p_SendRB, cs_PDCP_NoHeaderPDU(p_Data))));

    //@siclog "Step 15" siclog@
    //Receive the looped back IP packet    
    alt {      
      [] U_AM.receive(car_RLC_AM_TestData_Ind(cr_RLC_AM_DataInd(utran_CellDedicated, p_ReceiveRB, cr_PDCP_NoHeaderPDU(p_Data)))) 
        {
          t_Dly.stop;
        }
      [] t_Dly.timeout
        {
          if (p_Loopback) {
            f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Timeout");
          } else {
            f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "PDU was not looped back");
          }
        }
    }
  }


After:

	function f_UTRAN34_SendReceiveData(SS_RB_Identity p_SendRB,
                                     SS_RB_Identity p_ReceiveRB,
                                     bitstring p_Data,
                                     boolean p_Loopback := true) runs on UTRAN_PTC
  {
    timer t_Dly := 1.0; //1s
    //WA#NISPC_Generic : The loopback data should be sent by PDCP, not as RLC AM data
    //U_AM.send(cas_RLC_AM_TestData_Req(cs_RLC_AM_DataReq(utran_CellDedicated, p_SendRB, cs_PDCP_NoHeaderPDU(p_Data))));
     U_PDCP.send(cas_PDCP_Test_Data_Req (utran_CellDedicated,  //@sic R5s130195 PDCP sic@
                                     p_SendRB,
                                     p_Data));
    //@siclog "Step 15" siclog@
    //Receive the looped back IP packet
    t_Dly.start;//WA#NISPC_Generic
    alt {
      //WA#NISPC_Generic : The loopback data should be received by PDCP, not as RLC AM data Ind
      //[] U_AM.receive(car_RLC_AM_TestData_Ind(cr_RLC_AM_DataInd(utran_CellDedicated, p_ReceiveRB, cr_PDCP_NoHeaderPDU(p_Data))))
        [] U_PDCP.receive(car_PDCP_Test_Data_Ind (utran_CellDedicated,  //@sic R5s130195 PDCP sic@
                                        p_ReceiveRB,//WA#NISPC_Generic p_SendRB,
                                        p_Data))
        {
          t_Dly.stop;
        }
      [] t_Dly.timeout
        {
          if (p_Loopback) {
            f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Timeout");
          } else {
            f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "PDU was not looped back");
          }
        }
    }
  }


	//WA#NISPC_Generic : templates for PDCP data handling
  template (present) PDCP_DATA_IND car_PDCP_Test_Data_Ind(UTRAN_CellId_Type  p_CellId,
                                                          integer            p_RB_Id,
                                                          bitstring          p_Data) :=
  { /* @status    APPROVED (LTE_IRAT) */
    cellId := f_UtranCellId2Int(p_CellId),
      rB_Id := p_RB_Id,
    data := {
      data := p_Data
    }
  };
  //WA#NISPC_Generic : templates for PDCP data handling
  template (value) PDCP_DATA_REQ cas_PDCP_Test_Data_Req(UTRAN_CellId_Type          p_CellId,
                                                        integer                    p_RB_Id,
                                                        bitstring                  p_Data) :=
  { /* @status    APPROVED (LTE_IRAT) */
      cellId := f_UtranCellId2Int(p_CellId),
      rB_Id := p_RB_Id,
      data := {
        data := p_Data
      }
}


4.5 Function fl_TC_11_1_5_2_Body 
	Record Name
	fl_TC_11_1_5_2_Body

	Reason for change
	In the current TTCN implementation of function fl_TC_11_1_5_2_Body, for the Step 20 of the prose, the template cs_DeactPdpContextReqMt is being used to Deactivate the PDP Context. However, this template sets the tear down indicator to TRUE. This would result in the Deactivation of not just the Secondary PDP Context, but the Primary PDP context will be deactivated as well. This needs to be corrected. 
Also, the Traffic Flow Template that’s being generated before the Step 24 (Activate Secondary PDP context REQ followed by Activate Secondary PDP Context Accept) is incorrectly setting the Type of Service/Traffic class component and precedence. This neds to be corrected as well.

	Summary of change
	Created a new template cs_DeactPdpContextReqMt_NoTearDown and used it instead of cs_DeactPdpContextReqMt.
Corrected the Type of Service/Traffic class component and precedence before Step 24
Please see screenshot. 

	Source of change
	UTRAN34_RRC_Templates.ttcn

	
	


Before:

	U_Dc.send(cas_DataReq_PS(utran_CellDedicated, tsc_RB3, cs_U_DeactPdpContextReq(/*WA#11_1_5_2 cs_DeactPdpContextReqMt*/ cs_DeactPdpContextReqMt_NoTearDown(v_TI))));
    v_TI := v_SecPDPContextParameters.received_TI;
    fl_NISPC_WaitForPS_UplinkDirectTransfer(tsc_T3395max, cr_U_DeactPdpContextAcp(cr_DeactPdpContextAcpMo(v_TI)), "Test Case 11.1.5.2, Step 21");
    //@siclog "Step 21" siclog@
    //SS reconfigures RAB
    f_UTRAN34_ReleaseAM_RAB22_r7(utran34_Cell1);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 11.1.5.2, Step 21");
    //Generate traffic flow template #2
    //Same as packet filter#1, except for Type of Service/Traffic Class (TOS_TC) component and precedence
    v_TFT.packetFilterList[0].contents[0] := cs_PktFilterCompTOSTrafficClass('A0FC'O);    

    v_TFT.packetFilterList[0].precedence := int2bit(5, 8); 


After:

	U_Dc.send(cas_DataReq_PS(utran_CellDedicated, tsc_RB3, cs_U_DeactPdpContextReq(/*WA#11_1_5_2 cs_DeactPdpContextReqMt*/ cs_DeactPdpContextReqMt_NoTearDown(v_TI))));
    v_TI := v_SecPDPContextParameters.received_TI;
    fl_NISPC_WaitForPS_UplinkDirectTransfer(tsc_T3395max, cr_U_DeactPdpContextAcp(cr_DeactPdpContextAcpMo(v_TI)), "Test Case 11.1.5.2, Step 21");
    //@siclog "Step 21" siclog@
    //SS reconfigures RAB
    f_UTRAN34_ReleaseAM_RAB22_r7(utran34_Cell1);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 11.1.5.2, Step 21");
    //Generate traffic flow template #2
    //Same as packet filter#1, except for Type of Service/Traffic Class (TOS_TC) component and precedence
    //v_TFT.packetFilterList[0].contents[0] := cs_PktFilterCompTOSTrafficClass('A0FC'O);
    v_TFT.packetFilterList[0].contents[4] := cs_PktFilterCompTOSTrafficClass('A0FC'O);// WA#11_1_5_1 : Corrected position of TOS_TC in packet filter
    v_TFT.packetFilterList[0].precedence := int2bit(5, 8); 


	//WA#11_1_5_2 : New template for Deactivate PDP Context REQ without teardown indicator.

  template (value) DEACTIVATEPDPCONTEXTREQUEST cs_DeactPdpContextReqMt_NoTearDown(TI p_TI,

                                                                       CauseValue p_CauseValue := tsc_SM_CauseValue_RegularDeactivation) :=

  { /* @status    APPROVED (UTRAN) */

    ti                       := p_TI,

    sM_ProtocolDiscriminator := tsc_SMPD,

    msgType                  := '01000110'B,

    sM_Cause                 := cs_SM_Cause(p_CauseValue),

    tearDwnInd               := omit,

    protocolConfOpts         := omit,

    mBMS_ProtocolConfOpts    := omit,

    t3396Value               := omit // @sic R5s130195 Baseline Moving sic@

  };


4.6 Function f_UTRAN34_ReleaseAM_RAB22_r7
	Record Name
	f_UTRAN34_ReleaseAM_RAB22_r7

	Reason for change
	The function f_UTRAN34_ReleaseAM_RAB22_r7 is used to release the RB 22 after the UE transmits the Deactivate PDP Context Accept message. However, because of the local end mismatch, the SS does not receive the RB ReleaseComplete message. The local end configuration needs to be re-done for this function.

	Summary of change
	Corrected the implementation local end in the function f_UTRAN34_ReleaseAM_RAB22_r7. For this, created following new templates and used them as shown in the screenshot: cs_DCH_336_148_DL_Info_NISPC, cs_TrChInfoUL_336_148_FDD_NISPC, cs_TrChInfoDL_336_148_FDD_NISPC.
Also, the secondary scrambling code being set for the DL is incorrect, which has been corrected. 

	Source of change
	UTRAN34_RRC_Templates.ttcn

	
	


Before:

	function f_UTRAN34_ReleaseAM_RAB22_r7(UTRAN_CellId_Type p_CellId) runs on UTRAN_PTC
  {
    var UTRAN_SecurityInfo_Type v_SecurityInfo;
    var RRC_TransactionIdentifier v_RRC_Ti := tsc_RRC_TI;
    var UTRAN_CFN_Info_Type v_CFN_Info;
    var UL_ScramblingCode v_UL_ScrCode := f_UTRAN_CellInfo_GetULScrmCode_FDD(p_CellId);
    var PrimaryScramblingCode v_PrimScrCode := f_UTRAN_CellInfo_GetPriScrmCode_FDD(p_CellId);
    var FrequencyInfo v_FreqInfo := f_UTRAN_CellInfo_GetFrequencyInfo(p_CellId);
    v_CFN_Info := f_CalculateActTime(p_CellId);
    v_SecurityInfo := f_UTRAN_Security_Get();
    //@sic R5s140400 sic@
    U_AM.send(cas_RLC_Data_Req_Cnf(utran_CellDedicated, tsc_RB2, tsc_Mui, cs_RB_Rel_AM22_PS_r7(v_SecurityInfo.dl_IntegrityCheckInfo,
                                                                                               v_CFN_Info.actTime,
                                                                                               v_FreqInfo,
                                                                                               v_PrimScrCode,
                                                                                               v_UL_ScrCode)));
    U_AM.receive(car_AM_DataMuiCnf(utran_CellDedicated,tsc_RB2, tsc_Mui));  //@sic R5s140400 sic@
    //Reconfig Stand Alone SRB
    f_UTRAN_CRLC_Rel(utran_CellDedicated, tsc_RB22, UTRAN_FDD);   //@sic R5s140400 sic@
    f_UTRAN34_SS_2DCH_Modify_FDD_r7(p_CellId,
                                    cs_CphyTrchConfigReq_UL_DCH_336_148_FDD(cs_ActivateCFN(v_CFN_Info.actTime)),
                                    cs_CphyTrchConfigReq_DL_DCH_336_148_FDD(cs_ActivateCFN(v_CFN_Info.actTime)),                                 
                                    cs_TrChInfoUL_336_148_FDD,
                                    cs_TrChInfoDL_336_148_FDD,
                                    cs_TrLogMappingUL_4DCCH_1DTCH(tsc_RB20),
                                    cs_TrLogMappingDL_4DCCH_1DTCH_PS,
                                    v_CFN_Info.actTime,
                                    cs_DL_DPCH_64k_PS_FDDr7(cs_DL_CommonInformationRB_SetUp_FDD_r7(tsc_DL_DPCH1_SFP_64k_PS),
                                                            tsc_DL_DPCH_ScrC_3),
                                    cs_UL_DPCH_Info_FDDr7(tsc_UL_DPDCH_SF_64k_PS, pl0_96, v_UL_ScrCode));
    f_UTRAN_CellInfo_SetConfigType(p_CellId, cell_DCH_64kPS_RAB_SRB);
    //Receive RB Release Complete message
    U_AM.receive(car_RLC_Data_Ind(utran_CellDedicated, tsc_RB2, cr_108_radioBearerReleaseComplete(v_RRC_Ti, *, *)));
  }


After:

	function f_UTRAN34_ReleaseAM_RAB22_r7(UTRAN_CellId_Type p_CellId) runs on UTRAN_PTC
  {
    var UTRAN_SecurityInfo_Type v_SecurityInfo;
    var RRC_TransactionIdentifier v_RRC_Ti := tsc_RRC_TI;
    var UTRAN_CFN_Info_Type v_CFN_Info;
    var UL_ScramblingCode v_UL_ScrCode := f_UTRAN_CellInfo_GetULScrmCode_FDD(p_CellId);
    var PrimaryScramblingCode v_PrimScrCode := f_UTRAN_CellInfo_GetPriScrmCode_FDD(p_CellId);
    var FrequencyInfo v_FreqInfo := f_UTRAN_CellInfo_GetFrequencyInfo(p_CellId);
    v_CFN_Info := f_CalculateActTime(p_CellId);
    v_SecurityInfo := f_UTRAN_Security_Get();
    //@sic R5s140400 sic@
    U_AM.send(cas_RLC_Data_Req_Cnf(utran_CellDedicated, tsc_RB2, tsc_Mui, cs_RB_Rel_AM22_PS_r7(v_SecurityInfo.dl_IntegrityCheckInfo,
                                                                                               v_CFN_Info.actTime,
                                                                                               v_FreqInfo,
                                                                                               v_PrimScrCode,
                                                                                               v_UL_ScrCode)));
    U_AM.receive(car_AM_DataMuiCnf(utran_CellDedicated,tsc_RB2, tsc_Mui));  //@sic R5s140400 sic@
    //Reconfig Stand Alone SRB
    f_UTRAN_CRLC_Rel(utran_CellDedicated, tsc_RB22, UTRAN_FDD);   //@sic R5s140400 sic@
    f_UTRAN34_SS_2DCH_Modify_FDD_r7(p_CellId,
                                    cs_CphyTrchConfigReq_UL_DCH_336_148_FDD(cs_ActivateCFN(v_CFN_Info.actTime)),
                                    cs_DCH_336_148_DL_Info_NISPC(cs_ActivateCFN(v_CFN_Info.actTime)),//WA#11_1_5_2
                                    cs_TrChInfoUL_336_148_FDD_NISPC, //WA#11_1_5_2 : cs_TrChInfoUL_336_148_FDD,
                                    cs_TrChInfoDL_336_148_FDD_NISPC, //WA#11_1_5_2 : cs_TrChInfoUL_336_148_FDD,
                                    cs_TrLogMappingUL_4DCCH_1DTCH(tsc_RB20),
                                    cs_TrLogMappingDL_4DCCH_1DTCH_PS,
                                    v_CFN_Info.actTime,
                                    cs_DL_DPCH_64k_PS_FDDr7(cs_DL_CommonInformationRB_SetUp_FDD_r7(tsc_DL_DPCH1_SFP_64k_PS),
                                                            tsc_DL_DPCH1_2ndScrC/*WA#11_1_5_2 tsc_DL_DPCH_ScrC_3*/),
                                    cs_UL_DPCH_Info_FDDr7(tsc_UL_DPDCH_SF_64k_PS, pl0_96, v_UL_ScrCode));
    f_UTRAN_CellInfo_SetConfigType(p_CellId, cell_DCH_64kPS_RAB_SRB);
    //Receive RB Release Complete message
    U_AM.receive(car_RLC_Data_Ind(utran_CellDedicated, tsc_RB2, cr_108_radioBearerReleaseComplete(v_RRC_Ti, *, *)));
  }


	template (value) CphyTrchConfigReq cs_DCH_336_148_DL_Info_NISPC(template (value) SS_ActivationTime p_SS_ActTime) :=
  /* @descr For DCH1 (TS 34.108, cl 6.10.2.4.26):
   * TTI = 20 ms; 5 transport formats: TB size always=336 bits; Transport Blocks = 0,1,2,3,4; coding = turbo; CRCSize = 16; RateMatching = 130
   * For DCH5 (TS 34.108, cl 6.10.2.4.1.26):
   * TTI = 40 ms; 2 transport formats: TransportBlock = 1, TB size =148 bits; TransportBlock = 0, TB size = 148;
   * coding = convolutional; coding rate = 1/3; RateMatching = 155 */
  /* @status    APPROVED (UTRAN) */
    cs_CphyTrchConfigReq_DL_TrCh_FDD(p_SS_ActTime,
                                     cs_DCH_336_148_DL_Info_FDD,                                    cs_TFCS_Cmpl0_1_2_3_4_5_6_7_8_9_Tx(cs_PowerOffsetInfoHigher64k));
template (value) TrCHInfo cs_TrChInfoUL_336_148_FDD_NISPC := cs_TrCHInfo_UL_TrCh_FDD(cs_TrChInfoTrCHListUL_336_148_FDD, cs_TFCS_Cmpl0_1_2_3_4_5_6_7_8_9_Rx);                              /* @status    APPROVED (UTRAN) */

template (value) TrCHInfo cs_TrChInfoDL_336_148_FDD_NISPC := cs_TrCHInfo_DL_TrCh_FDD(cs_TrChInfoTrCHListDL_336_148_FDD, cs_TFCS_Cmpl0_1_2_3_4_5_6_7_8_9_Tx(cs_PowerOffsetInfoHigher64k)); /* @status    APPROVED (UTRAN) */


4.7 Function f_TC_11_1_5_2_UTRAN() 
	Record Name
	f_TC_11_1_5_2_UTRAN

	Reason for change
	In the current TTCN implementation of this function, the RRC connection is not released before starting the postamble. This needs to be implemented.

	Summary of change
	Called the function f_UTRAN34_RRC_ConnRel  before starting the postamble. 

please see the screenshots below.

	Source of change
	UTRAN34_NISPC.ttcn

	
	


Before:

	function f_TC_11_1_5_2_UTRAN() runs on UTRAN_PTC
  {
    //Initialise UTRAN cells according to 34.108
    f_UTRAN34_Init();
    // Cell1 started switched on with default power level
    f_UTRAN34_SS_CreateCellDCH(utran34_Cell1);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_Cell1);
    //Idle update to bring UE to Cell_DCH state to register and to release connection again
    f_UTRAN34_Preamble(utran34_Cell1);
    f_UTRAN_TestBody_Set(true);
    fl_TC_11_1_5_2_Body();
    f_UTRAN_TestBody_Set(false);
    //Open test loop and release RRC connection
    fl_OpenUE_TestLoop(utran34_Cell1,ps_domain, cell_Dch);
    //Release UTRAN Cell
    f_UTRAN34_Postamble(utran34_Cell1, U1_IDLE);
  }


After:

	function f_TC_11_1_5_2_UTRAN() runs on UTRAN_PTC
  {
    //Initialise UTRAN cells according to 34.108
    f_UTRAN34_Init();
    // Cell1 started switched on with default power level
    f_UTRAN34_SS_CreateCellDCH(utran34_Cell1);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_Cell1);
    //Idle update to bring UE to Cell_DCH state to register and to release connection again
    f_UTRAN34_Preamble(utran34_Cell1);
    f_UTRAN_TestBody_Set(true);
    fl_TC_11_1_5_2_Body();
    f_UTRAN_TestBody_Set(false);
    //Open test loop and release RRC connection
    fl_OpenUE_TestLoop(utran34_Cell1,ps_domain, cell_Dch);
    //WA#11_1_5_2 Release RRC Connection
    f_UTRAN34_RRC_ConnRel(utran34_Cell1, cell_Dch);
    //Release UTRAN Cell
    f_UTRAN34_Postamble(utran34_Cell1, U1_IDLE);
  }


5 Branches executed in test case 11.1.5.2
The test case implementation was executed with Integrity and ciphering enabled.

6 Execution Log Files

6.1 MediaTek K2 Handset
The MediaTek K2 Handset passed this test case on Rohde & Schwarz CMW500. The documentation below is enclosed as evidence of the successful test case run [1]:

Test Case Execution log file: 
Rohde-Schwarz\tc_11_1_5_2_MediaTek_K2.log 
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)
In the log provided, masking is applied to prevent any disclosure of confidential information.
In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file
7 References

